

## Lista D

Carolina Carvalho Silva - 10705933

### 1. Comparação sistemas linear e não-linear

Analisando o sistema de um reservatório composto por um único tanque, chegou-se à seguinte equação diferencial não-linear para expressar a altura do nível do fluido no reservatório :

$$\dot{h} = \left( -\sqrt{\frac{\rho g h}{R}} + Q_e \right) \frac{1}{S}$$

Já a equação linearizada que expressa a variação dessa altura é a seguinte :

$$\dot{x} = -\frac{1}{2S} \sqrt{\frac{\rho g}{Rh_o}} x + \frac{1}{S} u$$

$$y = +1x + 0u$$

Deve-se comparar a resolução do sistema do reservatório através dos modelos linear e não-linear, considerando que o reservatório se inicia no nível de 2m, mas com vazão de entrada nula. O gráfico obtido e que expressa a variação do nível do fluido no reservatório em função do tempo se encontra abaixo :




Figura 1 - Nível do reservatório

Percebe-se, que nesse caso, as duas soluções são muito aproximadas.

## 2. Circuito elétrico

Partindo do circuito elétrico abaixo e se utilizando do seu modelo matemático, deve-se obter o gráfico do comportamento da corrente.




Figura 2 - Circuito elétrico

Utilizando-se da Lei das Malhas de Kirchhoff, obtém-se :

$$e_i - \frac{1}{C} \int I dt - RI = 0$$

Considerando que a corrente equivale à derivada temporal da carga :

$$e_i - R\dot{y} - \frac{1}{C}y = 0$$

$$\Rightarrow \dot{y} + \frac{1}{RC}y = \frac{e_i}{R}$$

Para melhor comparar à equação do reservatório, assume-se  $y(0) = 0$  como condição inicial. Assim, obtém-se o gráfico a seguir, que representa a variação da tensão em função do tempo e é análogo ao da variação do nível do reservatório.




Figura 3 - Variação da tensão no circuito elétrico em função do tempo

### 3. Reservatório de dois tanques

Considere o seguinte reservatório :




Figura 4 - Reservatório de dois tanques

Fazendo uso do que foi calculado na Lista C, obtém-se as equações a seguir que expressam a altura dos níveis dos dois tanques :

$$\begin{cases} \dot{h}_1 = \left[ Q_e - \sqrt{\frac{\rho g}{R_1}(h_1 - h_2)} \right] \frac{1}{S_1} \\ \dot{h}_2 = \left[ \sqrt{\frac{\rho g}{R_1}(h_1 - h_2)} - \sqrt{\frac{\rho g}{R_1} h_2} \right] \frac{1}{S_2} \end{cases}$$

Para resolver o sistema linear, considera-se que :

$$A = \begin{bmatrix} -\frac{\rho g}{2SQ_e^0 R} & \frac{\rho g}{2SQ_e^0 R} \\ \frac{\rho g}{2SQ_e^0 R} & \frac{\rho g}{SQ_e^0 R} \end{bmatrix}; \quad B = \begin{bmatrix} \frac{1}{S} \\ 0 \end{bmatrix}; \quad C = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}; \quad D = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

Deve-se substituir essas matrizes de acordo com as equações :

$$\begin{cases} \dot{x} = Ax + Bu \\ y = Cx + Du \end{cases}$$

Assim, chega-se ao gráfico abaixo, que representa a altura do nível dos dois tanques em função do tempo :




Figura 5 - Altura do nível do fluido em dois tanques

#### 4. Código

##### a. Reservatório simples

```
clear all
xdel()

// Parametros:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
Qei=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada

// Sistema linear:
A=(-1/(2*S))*sqrt(rho*g/(R*ho));
B=1/S;
C=1;
D=0;
tanque=syslin('c',A,B,C,D);

// Condicao inicial
x0=0; // [m] desvio inicial do nivel em relação ao equilibrio
h0 = ho + x0; // [m] nivel inicial do tanque

// Instantes de tempo
t=0:10:40000;

// Vetor de entradas:
u=Qei*ones(t);

// Simulando o sistema como linear e plotando em azul
[y,x]=csim(u,t,tanque,x0);
plot2d(t,y,2)

Qei=sqrt(rho*g*(ho+hi)/R);
// Sistema não-linear:
funcprot(0);
function fun=tanque_n_lin(t, h)
fun = (-sqrt(rho*g*h/R)+Qei)*(1/S)
endfunction

// Simulando o sistema não-linear e plotando em vermelho
h=ode(h0,t(1),t,tanque_n_lin);
plot2d(t,h-ho,5)

xlabel("Resposta do tanque", "Tempo t [s]", "Variacao de nivel [m]");
legend(['Modelo linear'; 'Modelo não-linear']);
xgrid(4)
```

## b. Reservatório de dois tanques

```
clear all
xdel()

// Parametros:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=0.1; // [m] nivel adicional desejado
Qei=(1/2)*sqrt(rho*g/(R*(ho-hi))); // [m^3/s] vazao na entrada

a=(rho*g)/(2*S*Qei*R);
A = [-a,a;a,-2*a];
B = [1/S;0];
C = [1,0;0,1];
D = [0;0];

tanques=syslin('c',A,B,C,D);

//Condicoes iniciais
h0_1=1;
h0_2=4;
h0 = [h0_1;h0_2];

//Tempo
t=0:10:900;

//Entrada
u=Qei*ones(t);

[y,x]=csim(u,t,tanques,h0);
h1 = x(1,:);
h2 = x(2,:);

plot2d(t,h1,2)
plot2d(t,h2,3)

xtitle("Reservatório de dois tanques", "Tempo t [s]", "Altura do nível [m]");
legend(['Tanque 1'; 'Tanque 2']);

xgrid(4)
```